出欠席簿自動作成・計算VBA（2025年更新版）

Option Explicit

Private Const 開始行 As Integer = 3
Private Const 開始列 As Integer = 2
Private Const 名前列 As Integer = 2
Private Const 日付開始列 As Integer = 3

'**************************************************
' 設定保存用の構造体
'**************************************************
Private Type 簿設定
    児童数 As Integer
    出席記号 As String
    欠席記号 As String
    病欠記号 As String
    理由数 As Integer
    理由内容 As String
    理由記号 As String
End Type

Sub 出欠席簿作成()

    '前回値を記憶する変数（Static変数）
    Static 前回児童数 As Integer
    Static 前回出席記号 As String
    Static 前回欠席記号 As String
    Static 前回病欠記号 As String
    Static 前回理由数 As Integer
    Static 前回理由内容 As String
    Static 前回理由記号 As String
    
    '入力変数の定義
    Dim 児童数 As Integer
    Dim 月末日 As Integer
    Dim 土曜日 As String
    Dim 休園日 As String
    Dim 行事 As String
    Dim 理由数 As Integer
    Dim 理由内容() As String
    Dim 理由記号() As String
    Dim 出席記号 As String
    Dim 欠席記号 As String
    Dim 病欠記号 As String
    Dim 年 As Integer
    Dim 月 As Integer
    
    'その他の変数
    Dim i As Long
    Dim 設定 As 簿設定
    Dim 保存設定読込 As Boolean
    Dim コピーした児童名() As String
    Dim 児童名数 As Integer
    
    On Error GoTo エラー処理
    
    '============================================
    ' 0. 前月コピー確認
    '============================================
    Dim 前月コピー結果 As Integer
    前月コピー結果 = 前月コピー確認()
    
    If 前月コピー結果 = 1 Then  'コピー実行
        Call 前月からコピー実行(前回児童数, 前回出席記号, 前回欠席記号, 前回病欠記号, _
                              前回理由数, 前回理由内容, 前回理由記号, コピーした児童名, 児童名数)
    ElseIf 前月コピー結果 = -1 Then  'キャンセル
        Exit Sub
    End If
    '前月コピー結果 = 0 の場合は通常処理続行
    
    '============================================
    ' 1. 保存設定の読込確認
    '============================================
    保存設定読込 = False
    If 設定シート存在確認() Then
        If MsgBox("前回保存した設定を読み込みますか？", vbYesNo + vbQuestion, "設定読込") = vbYes Then
            If 設定読込(設定) Then
                前回児童数 = 設定.児童数
                前回出席記号 = 設定.出席記号
                前回欠席記号 = 設定.欠席記号
                前回病欠記号 = 設定.病欠記号
                前回理由数 = 設定.理由数
                前回理由内容 = 設定.理由内容
                前回理由記号 = 設定.理由記号
                保存設定読込 = True
            End If
        End If
    End If
    
    '============================================
    ' 2. 児童数の入力
    '============================================
    If 児童名数 > 0 Then
        児童数 = 児童名数
        MsgBox "前月からコピーした児童数：" & 児童数 & "人", vbInformation
    Else
        児童数 = 児童数入力(前回児童数)
        If 児童数 = 0 Then Exit Sub
    End If
    前回児童数 = 児童数
    
    '============================================
    ' 3. 年月入力と土曜日計算
    '============================================
    If 年月入力(年, 月) = False Then Exit Sub
    
    月末日 = Day(DateSerial(年, 月 + 1, 0))  '自動計算
    
    '土曜日の自動計算または手入力
    土曜日 = 土曜日入力改善版(年, 月, 月末日)
    If 土曜日 = "CANCEL" Then Exit Sub
    
    休園日 = 休園日入力(月末日)
    If 休園日 = "CANCEL" Then Exit Sub
    
    行事 = 行事入力(月末日)
    If 行事 = "CANCEL" Then Exit Sub
    
    '============================================
    ' 4. 出欠席記号の入力
    '============================================
    If 記号入力(出席記号, 欠席記号, 病欠記号, 前回出席記号, 前回欠席記号, 前回病欠記号) = False Then
        Exit Sub
    End If
    
    '============================================
    ' 5. その他の理由の入力
    '============================================
    If その他理由入力(理由数, 理由内容, 理由記号, 前回理由数, 前回理由内容, 前回理由記号) = False Then
        Exit Sub
    End If
    
    '============================================
    ' 6. 重複チェック
    '============================================
    If 日付重複チェック(土曜日, 休園日, 行事) = False Then
        If MsgBox("日付に重複があります。このまま続行しますか？", vbYesNo + vbQuestion) = vbNo Then
            Exit Sub
        End If
    End If
    
    If 記号重複確認(出席記号, 欠席記号, 病欠記号, 理由数, 理由記号) = False Then
        If MsgBox("記号に重複があります。このまま続行しますか？", vbYesNo + vbQuestion) = vbNo Then
            Exit Sub
        End If
    End If
    
    '============================================
    ' 7. プレビュー表示
    '============================================
    If 設定プレビュー表示(児童数, 年, 月, 月末日, 土曜日, 休園日, 行事, _
                        出席記号, 欠席記号, 病欠記号, 理由数, 理由内容, 理由記号) = False Then
        Exit Sub
    End If
    
    '============================================
    ' 8. シートの作成開始
    '============================================
    Application.ScreenUpdating = False
    
    Cells.Clear
    
    Call 使い方ガイド表示
    Call 名前欄作成(児童数)
    
    '前月コピーした児童名を書き込む
    If 児童名数 > 0 Then
        Call 児童名書き込み(コピーした児童名, 児童名数)
    End If
    
    Call 日付曜日作成(月末日, 土曜日, 年, 月)
    
    Columns(日付開始列).Resize(, 月末日).ColumnWidth = 6
    
    Call 背景色設定(児童数, 月末日, 土曜日, 休園日, 行事)
    Call ドロップダウン作成改善版(児童数, 月末日, 出席記号, 欠席記号, 病欠記号, 理由数, 理由記号)
    Call 集計欄作成(児童数, 月末日, 出席記号, 欠席記号, 病欠記号, 理由数, 理由記号, 理由内容)
    Call 罫線作成(児童数, 月末日, 理由数)
    Call 週区切り線追加(児童数, 月末日, 土曜日)
    Call シート保護設定(児童数, 月末日)
    
    Range("A1").Select
    
    Application.ScreenUpdating = True
    
    '============================================
    ' 9. 設定の保存確認
    '============================================
    If MsgBox("今回の設定を保存しますか？" & vbCrLf & _
             "（次回起動時に読み込めます）", vbYesNo + vbQuestion, "設定保存") = vbYes Then
        設定.児童数 = 児童数
        設定.出席記号 = 出席記号
        設定.欠席記号 = 欠席記号
        設定.病欠記号 = 病欠記号
        設定.理由数 = 理由数
        If 理由数 > 0 Then
            設定.理由内容 = Join(理由内容, ",")
            設定.理由記号 = Join(理由記号, ",")
        Else
            設定.理由内容 = ""
            設定.理由記号 = ""
        End If
        Call 設定保存(設定)
    End If
    
    MsgBox "出欠席簿の作成が完了しました！" & vbCrLf & vbCrLf & _
           "【作成内容】" & vbCrLf & _
           "・児童数：" & 児童数 & "人" & vbCrLf & _
           "・対象月：" & 年 & "年" & 月 & "月", vbInformation, "作成完了"
    
    Exit Sub
    
エラー処理:
    Application.ScreenUpdating = True
    MsgBox "エラーが発生しました。" & vbCrLf & vbCrLf & _
           "【エラー内容】" & vbCrLf & _
           Err.Description & vbCrLf & vbCrLf & _
           "もう一度最初からやり直してください。", vbCritical, "エラー"
End Sub

'**************************************************
' 前月コピー機能
'**************************************************
Private Function 前月コピー確認() As Integer
    '戻り値: 1=コピー実行, 0=新規作成, -1=キャンセル
    Dim シート数 As Integer
    
    前月コピー確認 = 0
    
    シート数 = ThisWorkbook.Worksheets.Count
    
    '設定シートを除外してカウント
    Dim i As Integer
    Dim データシート数 As Integer
    データシート数 = 0
    
    For i = 1 To シート数
        If ThisWorkbook.Worksheets(i).Name <> "設定" Then
            データシート数 = データシート数 + 1
        End If
    Next i
    
    If データシート数 >= 1 Then
        Dim 応答 As VbMsgBoxResult
        応答 = MsgBox("前月のデータから児童名や設定をコピーしますか？" & vbCrLf & vbCrLf & _
                     "【はい】前月データをコピーして作成" & vbCrLf & _
                     "【いいえ】新規作成", vbYesNoCancel + vbQuestion, "前月コピー")
        
        If 応答 = vbYes Then
            前月コピー確認 = 1
        ElseIf 応答 = vbNo Then
            前月コピー確認 = 0
        Else
            前月コピー確認 = -1
        End If
    End If
End Function

Private Sub 前月からコピー実行(ByRef 前回児童数 As Integer, ByRef 前回出席 As String, _
                            ByRef 前回欠席 As String, ByRef 前回病欠 As String, _
                            ByRef 前回理由数 As Integer, ByRef 前回理由内容 As String, _
                            ByRef 前回理由記号 As String, _
                            ByRef コピーした児童名() As String, ByRef 児童名数 As Integer)
    On Error GoTo エラー処理
    
    Dim コピー元シート As Worksheet
    Dim シート一覧 As String
    Dim i As Integer
    Dim 選択シート名 As String
    Dim シート番号 As Integer
    
    児童名数 = 0  '初期化
    
    'データシートの一覧作成
    シート一覧 = ""
    For i = 1 To ThisWorkbook.Worksheets.Count
        If ThisWorkbook.Worksheets(i).Name <> "設定" Then
            If シート一覧 = "" Then
                シート一覧 = i & ":" & ThisWorkbook.Worksheets(i).Name
            Else
                シート一覧 = シート一覧 & vbCrLf & i & ":" & ThisWorkbook.Worksheets(i).Name
            End If
        End If
    Next i
    
    'シート選択
    Dim 入力値 As String
    入力値 = InputBox("コピー元のシート番号を入力してください。" & vbCrLf & vbCrLf & _
                     "【シート一覧】" & vbCrLf & シート一覧, "コピー元シート選択", "1")
    
    If 入力値 = "" Then Exit Sub
    
    If Not IsNumeric(入力値) Then
        MsgBox "番号を入力してください。", vbExclamation
        Exit Sub
    End If
    
    シート番号 = CInt(入力値)
    If シート番号 < 1 Or シート番号 > ThisWorkbook.Worksheets.Count Then
        MsgBox "正しいシート番号を入力してください。", vbExclamation
        Exit Sub
    End If
    
    Set コピー元シート = ThisWorkbook.Worksheets(シート番号)
    
    '何をコピーするか選択
    Dim 児童名コピー As Boolean
    Dim 設定コピー As Boolean
    
    Dim 応答 As VbMsgBoxResult
    応答 = MsgBox("【" & コピー元シート.Name & "】から児童名をコピーしますか？", vbYesNoCancel + vbQuestion, "児童名コピー")
    
    If 応答 = vbCancel Then Exit Sub
    児童名コピー = (応答 = vbYes)
    
    応答 = MsgBox("【" & コピー元シート.Name & "】から設定（記号など）をコピーしますか？", vbYesNoCancel + vbQuestion, "設定コピー")
    
    If 応答 = vbCancel Then Exit Sub
    設定コピー = (応答 = vbYes)
    
    If Not 児童名コピー And Not 設定コピー Then
        MsgBox "コピーする項目が選択されていません。", vbInformation
        Exit Sub
    End If
    
    '児童名を配列に保存
    If 児童名コピー Then
        Dim 表開始行 As Integer
        表開始行 = 6
        
        Dim 最終行 As Long
        最終行 = コピー元シート.Cells(コピー元シート.Rows.Count, 名前列).End(xlUp).Row
        
        If 最終行 >= 表開始行 + 2 Then
            児童名数 = 最終行 - 表開始行 - 1
            
            '配列の初期化
            ReDim コピーした児童名(1 To 児童名数)
            
            '名前を配列に保存
            Dim j As Long
            Dim カウント As Integer
            カウント = 0
            For j = 表開始行 + 2 To 最終行
                If コピー元シート.Cells(j, 名前列).Value <> "" Then
                    カウント = カウント + 1
                    コピーした児童名(カウント) = コピー元シート.Cells(j, 名前列).Value
                End If
            Next j
            
            児童名数 = カウント
            前回児童数 = 児童名数
            
            MsgBox "児童名をコピーしました（" & 児童名数 & "人）", vbInformation
        End If
    End If
    
    '設定のコピー
    If 設定コピー Then
        Dim 設定シート As Worksheet
        On Error Resume Next
        Set 設定シート = ThisWorkbook.Worksheets("設定")
        On Error GoTo エラー処理
        
        If Not 設定シート Is Nothing Then
            If 設定シート.Cells(2, 2).Value <> "" Then
                前回出席 = 設定シート.Cells(2, 2).Value
                前回欠席 = 設定シート.Cells(3, 2).Value
                前回病欠 = 設定シート.Cells(4, 2).Value
                前回理由数 = 設定シート.Cells(5, 2).Value
                前回理由内容 = 設定シート.Cells(6, 2).Value
                前回理由記号 = 設定シート.Cells(7, 2).Value
                MsgBox "設定をコピーしました", vbInformation
            End If
        End If
    End If
    
    Exit Sub
    
エラー処理:
    MsgBox "コピー中にエラーが発生しました。" & vbCrLf & Err.Description, vbExclamation
End Sub

'児童名書き込みサブルーチン
Private Sub 児童名書き込み(児童名配列() As String, 児童名数 As Integer)
    Dim 表開始行 As Integer
    Dim i As Integer
    表開始行 = 6
    
    For i = 1 To 児童名数
        Cells(表開始行 + 1 + i, 名前列).Value = 児童名配列(i)
    Next i
End Sub

'============================================
' 入力関数群
'============================================

'児童数入力関数
Private Function 児童数入力(前回値 As Integer) As Integer
    Dim 入力値 As String
    Dim 数値 As Integer
    
    Do
        If 前回値 = 0 Then
            入力値 = InputBox( _
                "児童数を半角数字で入力してください。" & vbCrLf & vbCrLf & _
                "【入力例】30", _
                "児童数の入力", "30")
        Else
            If MsgBox( _
                "前回と同じ児童数（" & 前回値 & "人）で作成しますか？", _
                vbYesNo + vbQuestion, "児童数の確認") = vbYes Then
                児童数入力 = 前回値
                Exit Function
            Else
                入力値 = InputBox( _
                    "新しい児童数を半角数字で入力してください。" & vbCrLf & vbCrLf & _
                    "【入力例】30", _
                    "児童数の入力", CStr(前回値))
            End If
        End If
        
        If 入力値 = "" Then
            If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then
                児童数入力 = 0
                Exit Function
            End If
        ElseIf IsNumeric(入力値) Then
            数値 = CInt(入力値)
            If 数値 > 0 And 数値 <= 100 Then
                児童数入力 = 数値
                Exit Function
            Else
                MsgBox "1～100の範囲で入力してください。", vbExclamation
            End If
        Else
            MsgBox "半角数字で入力してください。", vbExclamation
        End If
    Loop
End Function

'年月入力関数
Private Function 年月入力(ByRef 年 As Integer, ByRef 月 As Integer) As Boolean
    Dim 入力値 As String
    Dim 数値 As Integer
    
    年月入力 = False
    
    '年の入力
    Do
        入力値 = InputBox( _
            "対象年を半角数字で入力してください。" & vbCrLf & vbCrLf & _
            "【入力例】2025", _
            "年の入力", CStr(Year(Date)))
        
        If 入力値 = "" Then
            If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then
                Exit Function
            End If
        ElseIf IsNumeric(入力値) Then
            数値 = CInt(入力値)
            If 数値 >= 2000 And 数値 <= 2100 Then
                年 = 数値
                Exit Do
            Else
                MsgBox "2000～2100の範囲で入力してください。", vbExclamation
            End If
        Else
            MsgBox "半角数字で入力してください。", vbExclamation
        End If
    Loop
    
    '月の入力
    Do
        入力値 = InputBox( _
            "対象月を半角数字で入力してください。" & vbCrLf & vbCrLf & _
            "【入力例】10", _
            "月の入力", CStr(Month(Date)))
        
        If 入力値 = "" Then
            If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then
                Exit Function
            End If
        ElseIf IsNumeric(入力値) Then
            数値 = CInt(入力値)
            If 数値 >= 1 And 数値 <= 12 Then
                月 = 数値
                Exit Do
            Else
                MsgBox "1～12の範囲で入力してください。", vbExclamation
            End If
        Else
            MsgBox "半角数字で入力してください。", vbExclamation
        End If
    Loop
    
    年月入力 = True
End Function

'土曜日入力改善版（自動計算オプション付き）
Private Function 土曜日入力改善版(年 As Integer, 月 As Integer, 月末日 As Integer) As String
    Dim 応答 As VbMsgBoxResult
    Dim 自動計算結果 As String
    
    応答 = MsgBox( _
        年 & "年" & 月 & "月の土曜日を自動計算しますか？" & vbCrLf & vbCrLf & _
        "【はい】カレンダーから自動計算" & vbCrLf & _
        "【いいえ】手動で入力", _
        vbYesNoCancel + vbQuestion, "土曜日の入力方法")
    
    If 応答 = vbCancel Then
        土曜日入力改善版 = "CANCEL"
        Exit Function
    End If
    
    If 応答 = vbYes Then
        '自動計算
        自動計算結果 = 土曜日自動計算(年, 月)
        
        If MsgBox( _
            "自動計算結果：" & 自動計算結果 & vbCrLf & vbCrLf & _
            "この内容でよろしいですか？", _
            vbYesNo + vbQuestion, "確認") = vbYes Then
            土曜日入力改善版 = 自動計算結果
        Else
            土曜日入力改善版 = 土曜日入力(月末日)
        End If
    Else
        '手動入力
        土曜日入力改善版 = 土曜日入力(月末日)
    End If
End Function

'土曜日自動計算関数
Private Function 土曜日自動計算(年 As Integer, 月 As Integer) As String
    Dim 日付 As Date
    Dim i As Integer
    Dim 土曜日リスト As String
    Dim 月末日 As Integer
    
    月末日 = Day(DateSerial(年, 月 + 1, 0))
    土曜日リスト = ""
    
    For i = 1 To 月末日
        日付 = DateSerial(年, 月, i)
        
        If Weekday(日付) = 7 Then  '7=土曜日
            If 土曜日リスト = "" Then
                土曜日リスト = CStr(i)
            Else
                土曜日リスト = 土曜日リスト & "," & CStr(i)
            End If
        End If
    Next i
    
    土曜日自動計算 = 土曜日リスト
End Function

'土曜日入力関数（元の関数）
Private Function 土曜日入力(月末日 As Integer) As String
    Dim 入力値 As String
    
    Do
        入力値 = InputBox( _
            "土曜日の日にちを半角数字とカンマで入力してください。" & vbCrLf & _
            "（小さい順に、カンマで区切って入力）" & vbCrLf & vbCrLf & _
            "【入力例】2,9,16,23,30", _
            "土曜日の入力", "7,14,21,28")
        
        If 入力値 = "" Then
            If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then
                土曜日入力 = "CANCEL"
                Exit Function
            End If
        ElseIf 日付リスト検証(入力値, 月末日) Then
            土曜日入力 = 入力値
            Exit Function
        Else
            MsgBox "正しい形式で入力してください。" & vbCrLf & _
                   "（例：2,9,16,23）", vbExclamation
        End If
    Loop
End Function

'休園日入力関数
Private Function 休園日入力(月末日 As Integer) As String
    Dim 入力値 As String
    
    Do
        入力値 = InputBox( _
            "休園日（日曜日・祝日・振替休日等）の日にちを入力してください。" & vbCrLf & _
            "（小さい順に、カンマで区切って入力）" & vbCrLf & vbCrLf & _
            "【入力例】1,8,15,22,29" & vbCrLf & vbCrLf & _
            "※休園日がない場合は、そのままOKボタンを押してください。", _
            "休園日の入力", "")
        
        If 入力値 = "" Then
            休園日入力 = ""
            Exit Function
        ElseIf 日付リスト検証(入力値, 月末日) Then
            休園日入力 = 入力値
            Exit Function
        Else
            If MsgBox("入力形式が正しくありません。" & vbCrLf & _
                     "作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then
                休園日入力 = "CANCEL"
                Exit Function
            End If
        End If
    Loop
End Function

'行事入力関数
Private Function 行事入力(月末日 As Integer) As String
    Dim 入力値 As String
    
    Do
        入力値 = InputBox( _
            "行事がある日にちを入力してください。" & vbCrLf & _
            "（小さい順に、カンマで区切って入力）" & vbCrLf & vbCrLf & _
            "【入力例】5,12,19,26" & vbCrLf & vbCrLf & _
            "※行事がない場合は、そのままOKボタンを押してください。", _
            "行事日の入力", "")
        
        If 入力値 = "" Then
            行事入力 = ""
            Exit Function
        ElseIf 日付リスト検証(入力値, 月末日) Then
            行事入力 = 入力値
            Exit Function
        Else
            If MsgBox("入力形式が正しくありません。" & vbCrLf & _
                     "作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then
                行事入力 = "CANCEL"
                Exit Function
            End If
        End If
    Loop
End Function

'日付リスト検証関数
Private Function 日付リスト検証(リスト As String, 月末日 As Integer) As Boolean
    Dim 配列 As Variant
    Dim i As Long
    Dim 数値 As Integer
    
    日付リスト検証 = False
    
    If リスト = "" Then
        日付リスト検証 = True
        Exit Function
    End If
    
    配列 = Split(リスト, ",")
    
    For i = LBound(配列) To UBound(配列)
        If Not IsNumeric(Trim(配列(i))) Then Exit Function
        数値 = CInt(Trim(配列(i)))
        If 数値 < 1 Or 数値 > 月末日 Then Exit Function
    Next i
    
    日付リスト検証 = True
End Function

'記号入力関数
Private Function 記号入力(ByRef 出席 As String, ByRef 欠席 As String, ByRef 病欠 As String, _
                         前回出席 As String, 前回欠席 As String, 前回病欠 As String) As Boolean
    Dim 入力値 As String
    
    記号入力 = False
    
    If 前回出席 <> "" Then
        If MsgBox( _
            "前回と同じ記号を使用しますか？" & vbCrLf & vbCrLf & _
            "【前回の記号】" & vbCrLf & _
            "・出席：" & 前回出席 & vbCrLf & _
            "・欠席：" & 前回欠席 & vbCrLf & _
            "・病欠：" & 前回病欠, _
            vbYesNo + vbQuestion, "記号の確認") = vbYes Then
            出席 = 前回出席
            欠席 = 前回欠席
            病欠 = 前回病欠
            記号入力 = True
            Exit Function
        End If
    End If
    
    '出席記号
    Do
        入力値 = InputBox( _
            "出席の記号を1～2文字で入力してください。" & vbCrLf & vbCrLf & _
            "【入力例】〇、●、出", _
            "出席記号の入力", "〇")
        
        If 入力値 = "" Then
            If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then Exit Function
        ElseIf Len(入力値) <= 2 Then
            出席 = 入力値
            Exit Do
        Else
            MsgBox "1～2文字で入力してください。", vbExclamation
        End If
    Loop
    
    '欠席記号
    Do
        入力値 = InputBox( _
            "欠席の記号を1～2文字で入力してください。" & vbCrLf & vbCrLf & _
            "【入力例】／、×、欠", _
            "欠席記号の入力", "／")
        
        If 入力値 = "" Then
            If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then Exit Function
        ElseIf Len(入力値) <= 2 Then
            欠席 = 入力値
            Exit Do
        Else
            MsgBox "1～2文字で入力してください。", vbExclamation
        End If
    Loop
    
    '病欠記号
    Do
        入力値 = InputBox( _
            "病欠の記号を1～2文字で入力してください。" & vbCrLf & vbCrLf & _
            "【入力例】△、▲、病", _
            "病欠記号の入力", "△")
        
        If 入力値 = "" Then
            If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then Exit Function
        ElseIf Len(入力値) <= 2 Then
            病欠 = 入力値
            Exit Do
        Else
            MsgBox "1～2文字で入力してください。", vbExclamation
        End If
    Loop
    
    記号入力 = True
End Function

'その他理由入力関数
Private Function その他理由入力(ByRef 理由数 As Integer, ByRef 内容() As String, ByRef 記号() As String, _
                              前回数 As Integer, 前回内容 As String, 前回記号 As String) As Boolean
    Dim 入力値 As String
    Dim i As Long
    Dim 数値 As Integer
    
    その他理由入力 = False
    
    If 前回数 > 0 Then
        If MsgBox( _
            "前回と同じ「その他の理由」を使用しますか？" & vbCrLf & vbCrLf & _
            "【前回の設定】" & vbCrLf & _
            "・理由数：" & 前回数 & "件" & vbCrLf & _
            "・内容：" & 前回内容 & vbCrLf & _
            "・記号：" & 前回記号, _
            vbYesNo + vbQuestion, "その他の理由の確認") = vbYes Then
            
            理由数 = 前回数
            Dim 内容配列 As Variant
            Dim 記号配列 As Variant
            内容配列 = Split(前回内容, ",")
            記号配列 = Split(前回記号, ",")
            
            ReDim 内容(1 To 理由数)
            ReDim 記号(1 To 理由数)
            
            For i = 1 To 理由数
                If i - 1 <= UBound(内容配列) Then
                    内容(i) = 内容配列(i - 1)
                End If
                If i - 1 <= UBound(記号配列) Then
                    記号(i) = 記号配列(i - 1)
                End If
            Next i
            
            その他理由入力 = True
            Exit Function
        End If
    End If
    
    Do
        入力値 = InputBox( _
            "出席・欠席・病欠以外に記録したい理由の数を入力してください。" & vbCrLf & vbCrLf & _
            "【入力例】" & vbCrLf & _
            "・遅刻、早退、出席停止の3種類 → 3" & vbCrLf & _
            "・理由を追加しない場合 → 0", _
            "その他の理由数", "0")
        
        If 入力値 = "" Then
            If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then Exit Function
        ElseIf IsNumeric(入力値) Then
            数値 = CInt(入力値)
            If 数値 >= 0 And 数値 <= 10 Then
                理由数 = 数値
                Exit Do
            Else
                MsgBox "0～10の範囲で入力してください。", vbExclamation
            End If
        Else
            MsgBox "半角数字で入力してください。", vbExclamation
        End If
    Loop
    
    If 理由数 = 0 Then
        ReDim 内容(0)
        ReDim 記号(0)
        その他理由入力 = True
        Exit Function
    End If
    
    ReDim 内容(1 To 理由数)
    ReDim 記号(1 To 理由数)
    
    For i = 1 To 理由数
        Do
            入力値 = InputBox( _
                i & "番目の理由の内容を入力してください。" & vbCrLf & vbCrLf & _
                "【入力例】" & vbCrLf & _
                "遅刻、早退、出席停止、忌引、休園 等", _
                "理由内容（" & i & "/" & 理由数 & "）", "")
            
            If 入力値 = "" Then
                If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then Exit Function
            Else
                内容(i) = 入力値
                Exit Do
            End If
        Loop
        
        Do
            入力値 = InputBox( _
                i & "番目の理由（" & 内容(i) & "）の記号を1～2文字で入力してください。" & vbCrLf & vbCrLf & _
                "【入力例】" & vbCrLf & _
                "遅刻→遅、早退→早、出席停止→停、忌引→忌", _
                "理由記号（" & i & "/" & 理由数 & "）", "")
            
            If 入力値 = "" Then
                If MsgBox("作成を中止しますか？", vbYesNo + vbQuestion) = vbYes Then Exit Function
            ElseIf Len(入力値) <= 2 Then
                記号(i) = 入力値
                Exit Do
            Else
                MsgBox "1～2文字で入力してください。", vbExclamation
            End If
        Loop
    Next i
    
    その他理由入力 = True
End Function

'============================================
' エラー防止関数群
'============================================

'日付重複チェック関数
Private Function 日付重複チェック(土曜日 As String, 休園日 As String, 行事 As String) As Boolean
    Dim 土曜配列 As Variant
    Dim 休園配列 As Variant
    Dim 行事配列 As Variant
    Dim i As Long
    
    日付重複チェック = True
    
    If 土曜日 = "" And 休園日 = "" And 行事 = "" Then Exit Function
    
    Dim すべての日付 As Object
    Set すべての日付 = CreateObject("Scripting.Dictionary")
    
    '土曜日を追加
    If 土曜日 <> "" Then
        土曜配列 = Split(土曜日, ",")
        For i = LBound(土曜配列) To UBound(土曜配列)
            Dim 日 As String
            日 = Trim(土曜配列(i))
            If すべての日付.exists(日) Then
                日付重複チェック = False
                Exit Function
            End If
            すべての日付.Add 日, "土曜"
        Next i
    End If
    
    '休園日を追加
    If 休園日 <> "" Then
        休園配列 = Split(休園日, ",")
        For i = LBound(休園配列) To UBound(休園配列)
            日 = Trim(休園配列(i))
            If すべての日付.exists(日) Then
                日付重複チェック = False
                Exit Function
            End If
            すべての日付.Add 日, "休園"
        Next i
    End If
    
    '行事を追加
    If 行事 <> "" Then
        行事配列 = Split(行事, ",")
        For i = LBound(行事配列) To UBound(行事配列)
            日 = Trim(行事配列(i))
            If すべての日付.exists(日) Then
                日付重複チェック = False
                Exit Function
            End If
            すべての日付.Add 日, "行事"
        Next i
    End If
End Function

'記号重複確認関数
Private Function 記号重複確認(出席 As String, 欠席 As String, 病欠 As String, _
                            理由数 As Integer, 理由記号() As String) As Boolean
    Dim すべての記号 As Object
    Set すべての記号 = CreateObject("Scripting.Dictionary")
    Dim i As Long
    
    記号重複確認 = True
    
    'すべての記号を追加してチェック
    If すべての記号.exists(出席) Then
        記号重複確認 = False
        Exit Function
    End If
    すべての記号.Add 出席, "出席"
    
    If すべての記号.exists(欠席) Then
        記号重複確認 = False
        Exit Function
    End If
    すべての記号.Add 欠席, "欠席"
    
    If すべての記号.exists(病欠) Then
        記号重複確認 = False
        Exit Function
    End If
    すべての記号.Add 病欠, "病欠"
    
    If 理由数 > 0 Then
        For i = 1 To 理由数
            If すべての記号.exists(理由記号(i)) Then
                記号重複確認 = False
                Exit Function
            End If
            すべての記号.Add 理由記号(i), "理由" & i
        Next i
    End If
End Function

'設定プレビュー表示関数
Private Function 設定プレビュー表示(児童数 As Integer, 年 As Integer, 月 As Integer, 月末日 As Integer, _
                                 土曜日 As String, 休園日 As String, 行事 As String, _
                                 出席 As String, 欠席 As String, 病欠 As String, _
                                 理由数 As Integer, 理由内容() As String, 理由記号() As String) As Boolean
    Dim メッセージ As String
    Dim i As Long
    
    メッセージ = "【設定内容の確認】" & vbCrLf & vbCrLf & _
                "◆ 基本情報" & vbCrLf & _
                "　対象月：" & 年 & "年" & 月 & "月（" & 月末日 & "日まで）" & vbCrLf & _
                "　児童数：" & 児童数 & "人" & vbCrLf & vbCrLf & _
                "◆ 特別日" & vbCrLf & _
                "　土曜日：" & 土曜日 & vbCrLf & _
                "　休園日：" & IIf(休園日 = "", "なし", 休園日) & vbCrLf & _
                "　行　事：" & IIf(行事 = "", "なし", 行事) & vbCrLf & vbCrLf & _
                "◆ 記号" & vbCrLf & _
                "　出席：" & 出席 & vbCrLf & _
                "　欠席：" & 欠席 & vbCrLf & _
                "　病欠：" & 病欠
    
    If 理由数 > 0 Then
        メッセージ = メッセージ & vbCrLf & vbCrLf & "◆ その他の理由（" & 理由数 & "件）"
        For i = 1 To 理由数
            メッセージ = メッセージ & vbCrLf & "　" & 理由内容(i) & "：" & 理由記号(i)
        Next i
    End If
    
    メッセージ = メッセージ & vbCrLf & vbCrLf & "この内容で作成しますか？"
    
    If MsgBox(メッセージ, vbYesNo + vbQuestion, "最終確認") = vbYes Then
        設定プレビュー表示 = True
    Else
        設定プレビュー表示 = False
    End If
End Function

'============================================
' 設定保存・読込関数群
'============================================

'設定シート存在確認
Private Function 設定シート存在確認() As Boolean
    Dim ws As Worksheet
    設定シート存在確認 = False
    
    On Error Resume Next
    Set ws = ThisWorkbook.Worksheets("設定")
    On Error GoTo 0
    
    If Not ws Is Nothing Then
        設定シート存在確認 = True
    End If
End Function
'設定保存
Private Sub 設定保存(設定 As 簿設定)
    On Error GoTo エラー処理
    
    Dim 設定シート As Worksheet
    
    '設定シートの取得または作成
    On Error Resume Next
    Set 設定シート = ThisWorkbook.Worksheets("設定")
    On Error GoTo エラー処理
    
    If 設定シート Is Nothing Then
        Set 設定シート = ThisWorkbook.Worksheets.Add(After:=ThisWorkbook.Worksheets(ThisWorkbook.Worksheets.Count))
        設定シート.Name = "設定"
        設定シート.Visible = xlSheetVeryHidden  '非表示
    End If
    
    '設定を書き込む
    With 設定シート
        .Cells.Clear
        .Cells(1, 1).Value = "出欠席簿設定"
        .Cells(2, 1).Value = "出席記号"
        .Cells(2, 2).Value = 設定.出席記号
        .Cells(3, 1).Value = "欠席記号"
        .Cells(3, 2).Value = 設定.欠席記号
        .Cells(4, 1).Value = "病欠記号"
        .Cells(4, 2).Value = 設定.病欠記号
        .Cells(5, 1).Value = "理由数"
        .Cells(5, 2).Value = 設定.理由数
        .Cells(6, 1).Value = "理由内容"
        .Cells(6, 2).Value = 設定.理由内容
        .Cells(7, 1).Value = "理由記号"
        .Cells(7, 2).Value = 設定.理由記号
        .Cells(8, 1).Value = "児童数"
        .Cells(8, 2).Value = 設定.児童数
    End With
    
    MsgBox "設定を保存しました。", vbInformation
    Exit Sub
    
エラー処理:
    MsgBox "設定の保存に失敗しました。" & vbCrLf & Err.Description, vbExclamation
End Sub

'設定読込
Private Function 設定読込(ByRef 設定 As 簿設定) As Boolean
    On Error GoTo エラー処理
    
    Dim 設定シート As Worksheet
    Set 設定シート = ThisWorkbook.Worksheets("設定")
    
    設定読込 = False
    
    With 設定シート
        If .Cells(2, 2).Value = "" Then Exit Function
        
        設定.出席記号 = .Cells(2, 2).Value
        設定.欠席記号 = .Cells(3, 2).Value
        設定.病欠記号 = .Cells(4, 2).Value
        設定.理由数 = .Cells(5, 2).Value
        設定.理由内容 = .Cells(6, 2).Value
        設定.理由記号 = .Cells(7, 2).Value
        設定.児童数 = .Cells(8, 2).Value
    End With
    
    設定読込 = True
    Exit Function
    
エラー処理:
    設定読込 = False
End Function

'============================================
' シート作成関数群
'============================================

'使い方ガイド表示
Private Sub 使い方ガイド表示()
    With Range("B1:Z1")
        .Merge
        .Value = "【 出欠席簿 使い方ガイド 】"
        .Font.Bold = True
        .Font.Size = 14
        .Font.Color = RGB(255, 255, 255)
        .Interior.Color = RGB(68, 114, 196)
        .HorizontalAlignment = xlCenter
        .VerticalAlignment = xlCenter
        .RowHeight = 28
    End With
    
    Dim ガイド背景色 As Long
    ガイド背景色 = RGB(237, 241, 250)
    
    With Range("B2:Z2")
        .Merge
        .Value = "◆ 次月作成：画面下の「＋（新しいシート）」をクリック → 「開発」タブまたは「表示」タブ → 「マクロ」→「実行」をクリック"
        .Font.Size = 10
        .Font.Bold = True
        .HorizontalAlignment = xlLeft
        .VerticalAlignment = xlCenter
        .RowHeight = 22
        .Interior.Color = ガイド背景色
        .WrapText = False
    End With
    
    With Range("B3:M3")
        .Merge
        .Value = "◆ 記号削除：セルを選択 → キーボードの「Delete」キーを押す"
        .Font.Size = 10
        .Font.Bold = True
        .HorizontalAlignment = xlLeft
        .VerticalAlignment = xlCenter
        .RowHeight = 22
        .Interior.Color = ガイド背景色
    End With
    
    With Range("N3:Z3")
        .Merge
        .Value = "◆ メモ追加：セルを右クリック → 「新しいメモ」を選択"
        .Font.Size = 10
        .Font.Bold = True
        .HorizontalAlignment = xlLeft
        .VerticalAlignment = xlCenter
        .Interior.Color = ガイド背景色
    End With
    
    With Range("B4:S4")
        .Merge
        .Value = "◆ 保存方法：「名前を付けて保存」→ ファイルの種類で「Excelマクロ有効ブック（*.xlsm）」を選択"
        .Font.Size = 10
        .Font.Bold = True
        .HorizontalAlignment = xlLeft
        .VerticalAlignment = xlCenter
        .RowHeight = 22
        .Interior.Color = ガイド背景色
    End With
    
    With Range("T4:Z4")
        .Merge
        .Value = "◆ エラー時：「終了」をクリック → 再度マクロを実行"
        .Font.Size = 10
        .Font.Color = RGB(192, 0, 0)
        .Font.Bold = True
        .HorizontalAlignment = xlLeft
        .VerticalAlignment = xlCenter
        .Interior.Color = ガイド背景色
    End With
    
    With Range("B1:Z4")
        .Borders.LineStyle = xlContinuous
        .Borders.Weight = xlMedium
        .Borders.Color = RGB(68, 114, 196)
    End With
    
    Rows("5:5").RowHeight = 5
End Sub

'名前欄作成
Private Sub 名前欄作成(児童数 As Integer)
    Dim 表開始行 As Integer
    表開始行 = 6
    
    With Range(Cells(表開始行, 名前列), Cells(表開始行 + 1, 名前列))
        .Merge
        .Value = "名前"
        .HorizontalAlignment = xlCenter
        .VerticalAlignment = xlCenter
        .Font.Bold = True
        .Font.Size = 11
        .Borders(xlEdgeLeft).Weight = xlThick
        .Borders(xlEdgeRight).Weight = xlThick
        .Borders(xlEdgeTop).Weight = xlThick
        .Borders(xlEdgeBottom).Weight = xlThick
    End With
    
    With Range(Cells(表開始行 + 2, 名前列), Cells(表開始行 + 1 + 児童数, 名前列))
        .Borders(xlEdgeLeft).Weight = xlThick
        .Borders(xlEdgeRight).Weight = xlThick
        .Borders(xlInsideHorizontal).LineStyle = xlContinuous
    End With
End Sub

'日付曜日作成（年月情報を使用）
Private Sub 日付曜日作成(月末日 As Integer, 土曜日文字列 As String, 年 As Integer, 月 As Integer)
    Dim i As Long
    Dim 日付 As Date
    Dim 曜日名 As String
    Dim 表開始行 As Integer
    表開始行 = 6
    
    With Range(Cells(表開始行, 日付開始列), Cells(表開始行 + 1, 日付開始列 + 月末日 - 1))
        .Borders(xlEdgeTop).Weight = xlThick
        .Borders(xlEdgeBottom).Weight = xlThick
    End With
    
    For i = 1 To 月末日
        '日付
        With Cells(表開始行, 日付開始列 + i - 1)
            .Value = i
            .HorizontalAlignment = xlCenter
            .Font.Bold = True
        End With
        
        '曜日（DateSerial関数で正確に計算）
        日付 = DateSerial(年, 月, i)
        曜日名 = Format(日付, "aaa")  '日、月、火、水、木、金、土
        
        With Cells(表開始行 + 1, 日付開始列 + i - 1)
            .Value = 曜日名
            .HorizontalAlignment = xlCenter
        End With
    Next i
End Sub

'背景色設定
Private Sub 背景色設定(児童数 As Integer, 月末日 As Integer, _
                      土曜日 As String, 休園日 As String, 行事 As String)
    Dim i As Long
    Dim 配列 As Variant
    Dim 列番号 As Integer
    Dim 表開始行 As Integer
    表開始行 = 6
    
    '土曜日（水色）
    If 土曜日 <> "" Then
        配列 = Split(土曜日, ",")
        For i = LBound(配列) To UBound(配列)
            If IsNumeric(Trim(配列(i))) Then
                列番号 = CInt(Trim(配列(i))) + 日付開始列 - 1
                If 列番号 >= 日付開始列 And 列番号 < 日付開始列 + 月末日 Then
                    Range(Cells(表開始行, 列番号), Cells(表開始行 + 児童数 + 1, 列番号)).Interior.Color = RGB(204, 255, 255)
                End If
            End If
        Next i
    End If
    
    '休園日（ピンク）
    If 休園日 <> "" Then
        配列 = Split(休園日, ",")
        For i = LBound(配列) To UBound(配列)
            If IsNumeric(Trim(配列(i))) Then
                列番号 = CInt(Trim(配列(i))) + 日付開始列 - 1
                If 列番号 >= 日付開始列 And 列番号 < 日付開始列 + 月末日 Then
                    Range(Cells(表開始行, 列番号), Cells(表開始行 + 児童数 + 1, 列番号)).Interior.Color = RGB(255, 153, 204)
                End If
            End If
        Next i
    End If
    
    '行事（薄緑）
    If 行事 <> "" Then
        配列 = Split(行事, ",")
        For i = LBound(配列) To UBound(配列)
            If IsNumeric(Trim(配列(i))) Then
                列番号 = CInt(Trim(配列(i))) + 日付開始列 - 1
                If 列番号 >= 日付開始列 And 列番号 < 日付開始列 + 月末日 Then
                    Range(Cells(表開始行, 列番号), Cells(表開始行 + 児童数 + 1, 列番号)).Interior.Color = RGB(204, 255, 204)
                End If
            End If
        Next i
    End If
End Sub

'ドロップダウン作成（一括設定で高速化）
Private Sub ドロップダウン作成改善版(児童数 As Integer, 月末日 As Integer, _
                                 出席 As String, 欠席 As String, 病欠 As String, _
                                 理由数 As Integer, 理由記号() As String)
    Dim i As Long
    Dim 選択リスト As String
    Dim 表開始行 As Integer
    Dim 範囲 As Range
    表開始行 = 6
    
    選択リスト = 出席 & "," & 欠席 & "," & 病欠
    
    If 理由数 > 0 Then
        For i = 1 To 理由数
            選択リスト = 選択リスト & "," & 理由記号(i)
        Next i
    End If
    
    '一括設定で高速化
    Set 範囲 = Range(Cells(表開始行 + 2, 日付開始列), _
                    Cells(表開始行 + 児童数 + 1, 日付開始列 + 月末日 - 1))
    
    With 範囲
        .Validation.Delete
        .Validation.Add Type:=xlValidateList, _
                       AlertStyle:=xlValidAlertStop, _
                       Operator:=xlBetween, _
                       Formula1:=選択リスト
        .Validation.IgnoreBlank = True
        .Validation.InCellDropdown = True
        .HorizontalAlignment = xlCenter
    End With
End Sub

'集計欄作成
Private Sub 集計欄作成(児童数 As Integer, 月末日 As Integer, _
                      出席 As String, 欠席 As String, 病欠 As String, _
                      理由数 As Integer, 理由記号() As String, 理由内容() As String)
    Dim i As Long, k As Long
    Dim 最終列 As Integer
    Dim 開始列番号 As Integer
    Dim 表開始行 As Integer
    Dim 最大文字数 As Integer
    Dim 列幅 As Double
    表開始行 = 6
    
    最終列 = 日付開始列 + 月末日
    開始列番号 = 日付開始列
    
    '出席数
    With Range(Cells(表開始行, 最終列), Cells(表開始行 + 1, 最終列))
        .Merge
        .Value = "出席数"
        .HorizontalAlignment = xlCenter
        .VerticalAlignment = xlCenter
        .Font.Bold = True
        .Interior.Color = RGB(220, 230, 241)
    End With
    
    '欠席数
    With Range(Cells(表開始行, 最終列 + 1), Cells(表開始行 + 1, 最終列 + 1))
        .Merge
        .Value = "欠席数"
        .HorizontalAlignment = xlCenter
        .VerticalAlignment = xlCenter
        .Font.Bold = True
        .Interior.Color = RGB(220, 230, 241)
    End With
    
    '病欠数
    With Range(Cells(表開始行, 最終列 + 2), Cells(表開始行 + 1, 最終列 + 2))
        .Merge
        .Value = "病欠数"
        .HorizontalAlignment = xlCenter
        .VerticalAlignment = xlCenter
        .Font.Bold = True
        .Interior.Color = RGB(220, 230, 241)
    End With
    
    最大文字数 = 3
    
    If 理由数 > 0 Then
        For i = 1 To 理由数
            With Range(Cells(表開始行, 最終列 + 2 + i), Cells(表開始行 + 1, 最終列 + 2 + i))
                .Merge
                .Value = 理由内容(i) & "数"
                .HorizontalAlignment = xlCenter
                .VerticalAlignment = xlCenter
                .Font.Bold = True
                .Interior.Color = RGB(220, 230, 241)
                .WrapText = False
            End With
            
            If Len(理由内容(i)) > 最大文字数 Then
                最大文字数 = Len(理由内容(i))
            End If
        Next i
    End If
    
    If 最大文字数 <= 3 Then
        列幅 = 8
    ElseIf 最大文字数 = 4 Then
        列幅 = 11
    Else
        列幅 = 13
    End If
    
    Columns(最終列).Resize(, 3 + 理由数).ColumnWidth = 列幅
    
    '各児童の集計式
    For i = 表開始行 + 2 To 表開始行 + 児童数 + 1
        Cells(i, 最終列).Formula = "=COUNTIF(" & _
            Range(Cells(i, 開始列番号), Cells(i, 開始列番号 + 月末日 - 1)).Address & _
            ",""" & 出席 & """)"
        
        Cells(i, 最終列 + 1).Formula = "=COUNTIF(" & _
            Range(Cells(i, 開始列番号), Cells(i, 開始列番号 + 月末日 - 1)).Address & _
            ",""" & 欠席 & """)"
        
        Cells(i, 最終列 + 2).Formula = "=COUNTIF(" & _
            Range(Cells(i, 開始列番号), Cells(i, 開始列番号 + 月末日 - 1)).Address & _
            ",""" & 病欠 & """)"
        
        If 理由数 > 0 Then
            For k = 1 To 理由数
                Cells(i, 最終列 + 2 + k).Formula = "=COUNTIF(" & _
                    Range(Cells(i, 開始列番号), Cells(i, 開始列番号 + 月末日 - 1)).Address & _
                    ",""" & 理由記号(k) & """)"
            Next k
        End If
    Next i
    
    With Range(Cells(表開始行 + 2, 最終列), Cells(表開始行 + 児童数 + 1, 最終列 + 2 + 理由数))
        .HorizontalAlignment = xlCenter
        .NumberFormatLocal = "0"
    End With
End Sub

'罫線作成
Private Sub 罫線作成(児童数 As Integer, 月末日 As Integer, 理由数 As Integer)
    Dim 最終列 As Integer
    Dim 表開始行 As Integer
    Dim 月末列 As Integer
    表開始行 = 6
    
    最終列 = 日付開始列 + 月末日 + 2 + 理由数
    月末列 = 日付開始列 + 月末日 - 1
    
    With Range(Cells(表開始行, 名前列), Cells(表開始行 + 児童数 + 1, 最終列))
        .Borders(xlInsideVertical).LineStyle = xlContinuous
        .Borders(xlInsideVertical).Weight = xlThin
        .Borders(xlInsideHorizontal).LineStyle = xlContinuous
        .Borders(xlInsideHorizontal).Weight = xlThin
        
        .Borders(xlEdgeTop).LineStyle = xlContinuous
        .Borders(xlEdgeTop).Weight = xlMedium
        .Borders(xlEdgeBottom).LineStyle = xlContinuous
        .Borders(xlEdgeBottom).Weight = xlMedium
        .Borders(xlEdgeLeft).LineStyle = xlContinuous
        .Borders(xlEdgeLeft).Weight = xlMedium
        .Borders(xlEdgeRight).LineStyle = xlContinuous
        .Borders(xlEdgeRight).Weight = xlMedium
    End With
    
    Range(Cells(表開始行, 名前列 + 1), Cells(表開始行 + 児童数 + 1, 名前列 + 1)) _
        .Borders(xlEdgeLeft).Weight = xlMedium
    
    Range(Cells(表開始行, 月末列), Cells(表開始行 + 児童数 + 1, 月末列)) _
        .Borders(xlEdgeRight).Weight = xlThick
    Range(Cells(表開始行, 月末列), Cells(表開始行 + 児童数 + 1, 月末列)) _
        .Borders(xlEdgeRight).Color = RGB(0, 0, 0)
End Sub

'週区切り線追加
Private Sub 週区切り線追加(児童数 As Integer, 月末日 As Integer, 土曜日 As String)
    Dim 配列 As Variant
    Dim i As Long
    Dim 列番号 As Integer
    Dim 表開始行 As Integer
    表開始行 = 6
    
    If 土曜日 = "" Then Exit Sub
    
    配列 = Split(土曜日, ",")
    
    For i = LBound(配列) To UBound(配列)
        If IsNumeric(Trim(配列(i))) Then
            列番号 = CInt(Trim(配列(i))) + 日付開始列 - 1
            If 列番号 >= 日付開始列 And 列番号 < 日付開始列 + 月末日 Then
                '土曜日の右側に太線
                With Range(Cells(表開始行, 列番号), Cells(表開始行 + 児童数 + 1, 列番号))
                    .Borders(xlEdgeRight).Weight = xlMedium
                    .Borders(xlEdgeRight).Color = RGB(100, 100, 100)
                End With
            End If
        End If
    Next i
End Sub

'シート保護設定
Private Sub シート保護設定(児童数 As Integer, 月末日 As Integer)
    Dim 表開始行 As Integer
    表開始行 = 6
    
    On Error Resume Next
    
    'すべてのセルをロック
    Cells.Locked = True
    
    '名前欄と出欠欄のみロック解除
    Range(Cells(表開始行 + 2, 名前列), Cells(表開始行 + 児童数 + 1, 名前列)).Locked = False
    Range(Cells(表開始行 + 2, 日付開始列), Cells(表開始行 + 児童数 + 1, 日付開始列 + 月末日 - 1)).Locked = False
    
    'シートを保護（パスワードなし、VBAからの操作は許可）
    ActiveSheet.Protect Password:="", _
                        DrawingObjects:=True, _
                        Contents:=True, _
                        Scenarios:=True, _
                        UserInterfaceOnly:=True
    
    ActiveSheet.EnableSelection = xlUnlockedCells
    
    On Error GoTo 0
End Sub

